wxMAXIMA*

Δημήτριος Χριστόπουλος[1,2]

[1]Εθνικό Καποδιστριακό Πανεπιστήμιο Αθηνών, Τμήμα Οικονομικών Επιστημών
[2]dchristop@econ.uoa.gr

Άνοιξη 2011

Μία πολύ σύντομη Εισαγωγή στο wxMaxima

*Πρόγραμμα ανοιχτού λογισμικού με άδεια GNU General Public License, version 2 (GPLv2), διαθέσιμο: http://wxmaxima.sourceforge.net/.

Περιεχόμενα

Κατάλογος Σχημάτων

1 Εισαγωγή

Οι σημειώσεις αυτές γράφτηκαν για να καλύψουν την ανάγκη των φοιτητών του μαθήματος 'Γραμμικά Μαθηματικά' για ένα δωρεάν σύστημα υπολογιστικής άλγεβρας το οποίο θα τους επιτρέπει να κάνουν συμβολικές πράξεις της μορφής:

$$A^{\nu} = \begin{pmatrix} 1 & 2\nu & 4\nu - \nu^2 \\ 0 & 1 & -\nu \\ 0 & 0 & 1 \end{pmatrix}$$

Επιλέχθηκε, ανάμεσα σε πολλές υποψηφιότητες είναι αλήθεια, το wxMaxima πιο πολύ για λόγους ηθικής τάξης: όλα τα εμπορικά προγράμματα που διαδέχθηκαν το DOE Macsyma[1] είναι αυτά που χρησιμοποιούνται από την ακαδημαϊκή κοινότητα σήμερα, χωρίς όμως να είναι αρκετά γνωστό ότι βασίστηκαν στο ανωτέρω πρόγραμμα συμβολικής άλγεβρας. Ο άλλος λόγος για τον οποίο το επιλέξαμε είναι διότι είναι αρκετά φιλικό με τους χρήστες των Windows και έτσι μπορεί να χρησιμοποιηθεί από περισσότερους φοιτητές.
Πρόκειται λοιπόν για 'ελεύθερο λογισμικό' ή αλλιώς 'λογισμικό ανοιχτού κώδικα', το οποίο εξακολουθεί να εργάζεται με την γλώσσα συμβολικής άλγεβρας LISP ακόμα και σήμερα. Μάλιστα με την χρήση κατάλληλης εντολής μπορούμε να δούμε ακριβώς πως μεταφράζεται η κάθε εντολή μας στην ανωτέρω γλώσσα προγραμματισμού.

[1] Λογισμικό που αναπτύχθηκε στο M.I.T. την δεκαετία του '60 και κατέστη ελεύθερο λογισμικό το 1998 από τον Bill Schelter.

2 Ορισμοί και στοιχειώδεις πράξεις

Δίνουμε τιμή σε μία μεταβλητή με το σύμβολο : και αμέσως τοποθετείται ένα ερωτηματικό ; στο τέλος, πού σημαίνει ότι θα εμφανιστεί η τιμή στην οθόνη. Εάν δεν θέλουμε να εμφανιστεί η τιμή τότε βάζουμε το σύμβολο $ στο τέλος.
Υπάρχουν οι εντολές 'εισόδου' input, (%i1) και οι εντολές 'εξόδου' output, (%o1) με αρίθμηση για κάθε αρχείο. Εάν θέλουμε να ανακαλέσουμε την τελευταία εντολή πληκτρολογούμε τον χαρακτήρα % και εμφανίζεται αμέσως:

```
(%i1) a:1;
(%o1) 1

(%i2) b:2;
(%o2) 2

(%i3) a;
(%o3) 1

(%i4) b;
(%o4) 2

(%i5) %;
(%o5) 2
```

Μία λίστα εισάγεται με αγκύλες και κόμματα ανάμεσα στα στοιχεία της. Ας κάνουμε στοιχειώδεις πράξεις με τις μεταβλητές a,b που έχουμε ήδη ορίσει:

```
(%i10) [a+b,a-b,a*b,a/b,a^b];
```

(%o10) $[3, -1, 2, \frac{1}{2}, 1]$

Μπορούμε πατώντας το κεντρικό enter του πληκτρολογίου να κάνουμε πολλαπλές εισαγωγές ορισμάτων και στο τέλος με shift + enter ή με το enter του αριθμητικού πληκτρολογίου να ολοκληρωθεί η εισαγωγή και να εκτελεστούν οι εντολές όλες μαζί. Επίσης εάν προσέξετε το τριγωνάκι δίπλα σε κάθε ομάδα εντολών και το πατήσετε θα γίνει απόκρυψη των γραμμών εντολών που αυτό έχει. Το wxMaxima δέχεται ελληνικούς χαρακτήρες για ορίσματα.



(α') Εμφάνιση (β') Απόκρυψη



Όταν ανοίξουμε μία αριστερή παρένθεση τότε το wxMaxima αυτόματα εμφανίζει και την δεξιά παρένθεση, ώστε να μας βοηθήσει να μην την ξεχάσουμε.



(γ') Παρένθεση (δ') Συμπλήρωση κι αποτέλεσμα

3 Λύση εξισώσεων

Έστω η εξίσωση:

$$x^2 - 5x + 6 = 0 \tag{1}$$

Η λύση γίνεται με την εντολή solve. Χρησιμοποιήσαμε και την εντολή “kill all” για να κάνουμε εκκαθάριση της μνήμης.

```
(%i61) kill(all);
(%o0) done

(%i1) eq:x^2-5*x+6=0;
(%o1) x^2 -5 x+6=0

(%i2) solve(eq,x);
(%o2) [x=3, x=2]
```

Σχήμα 1: Λύση της εξίσωσης $x^2 - 5x + 6 = 0$

```
(%i1) x^3-x^2+4*x-4;
(%o1) x^3 -x^2 +4 x-4

(%i2) solve(%,x);
(%o2) [x=-2 %i, x=2 %i, x=1]
```

Σχήμα 2: Λύση της εξίσωσης $x^3 - x^2 + 4\,x - 4$

Ο χαρακτήρας %i είναι η φανταστική μονάδα, $i^2 = -1$.
Επίσης λύνουμε συστήματα οσωνδήποτε εξισώσεων με οσουσδήποτε αγνώστους.

```
(%i13) sys:[x+y+z=10,x-3*y+2*z=20,x-y+z=1];
(%o13) [z+y+x=10,2 z-3 y+x=20,z-y+x=1]
```

```
(%i14) solve(sys,[x,y,z]);
```
(%o14) $[[x=-\frac{45}{2}, y=\frac{9}{2}, z=28]]$

Σχήμα 3: Λύση συστήματος 3×3

4 Παραγώγιση

Μπορούμε να παραγωγίσουμε (**diff**erentiate) μία παράσταση p ως προς κάποια μεταβλητή x με την εντολή diff(p,x). Εάν πριν από κάποια εντολή θέσουμε το σύμβολο του ευθέος τόνου $'$, τότε η εντολή δεν εκτελείται, αλλά παρουσιάζεται μόνον.

```
(%i9) p:x^2+x*y+2*x^3+8*x^5*y+19;
```
(%o9) $8x^5y+xy+2x^3+x^2+19$

```
(%i10) diff(p,x);
```
(%o10) $40x^4y+y+6x^2+2x$

```
(%i11) diff(p,y);
```
(%o11) $8x^5+x$

```
(%i13) 'diff(p,x)=diff(p,x);
```
(%o13) $\frac{d}{dx}\left(8x^5y+xy+2x^3+x^2+19\right)=40x^4y+y+6x^2+2x$

```
(%i14) 'diff(p,y)=diff(p,y);
```
(%o14) $\frac{d}{dy}\left(8x^5y+xy+2x^3+x^2+19\right)=8x^5+x$

Σχήμα 4: Παραγώγιση

Το ολικό διαφορικό μίας συνάρτησης p πολλών μεταβλητών μπορεί να βρεθεί απλά με την εντολή diff(p), ενώ τα επιμέρους διαφορικά $dx, dy, \ldots$ παρουσιάζονται ως $del(x), del(y), \cdots$. Μπορούμε να ορίσουμε μία συνάρτηση πολλών μεταβλητών με τον τελεστή $:=$ να ακολουθείται από την έκφραση που δίνει τον τύπο της συνάρτησης. Εάν θέλουμε απλά το διαφορικό γράφουμε `diff(f)`, εάν θέλουμε να γίνουν οι μερικές παραγωγίσεις γράφουμε και το όρισμα της συνάρτησης, δηλ. `diff(f(x,y,z))`.

```
(%i25) kill(all);
(%o0)  done

(%i1)  f(x,y,z):=x^2+y^2+z^2-1;
(%o1)  f(x,y,z):=x^2+y^2+z^2-1

(%i2)  f(1,0,1);
(%o2)  1

(%i3)  diff(f(x,y,z));
(%o3)  2 z del(z)+2 y del(y)+2 x del(x)

(%i9)  diff(f)='diff(f(x,y,z));
(%o9)  del(f)=del(z^2+y^2+x^2-1)

(%i10) diff(f)=diff(f(x,y,z));
(%o10) del(f)=2 z del(z)+2 y del(y)+2 x del(x)
```

Σχήμα 5: Διαφορικό συνάρτησης πολλών μεταβλητών

Μπορούμε να υπολογίσουμε εύκολα τις *μερικές ελαστικότητες (partial elasticities)* της f ως προς x, y, z.

(%i9) `epsilon[f,x]=x/f(x,y,z)*diff(f(x,y,z),x);`

(%o9) $$\varepsilon_{f,x}=\frac{2\,x^2}{z^2+y^2+x^2-1}$$

(%i10) `epsilon[f,y]=y/f(x,y,z)*diff(f(x,y,z),y);`

(%o10) $$\varepsilon_{f,y}=\frac{2\,y^2}{z^2+y^2+x^2-1}$$

(%i11) `epsilon[f,z]=z/f(x,y,z)*diff(f(x,y,z),z);`

(%o11) $$\varepsilon_{f,z}=\frac{2\,z^2}{z^2+y^2+x^2-1}$$

Σχήμα 6: Υπολογισμός μερικών ελαστικοτήτων συνάρτησης πολλών μεταβλητών

Μπορούμε να κάνουμε σύνθετη παραγώγιση. Ας υποθέσουμε ότι στην ίδια συνάρτηση ισχύει $x(t) = t^2 + 2\,s^3, y(t) = t^3 - 5\,s^7, z(t) = t^4 + 6\,s^2$. Τότε μπορούμε να υπολογίσουμε το ολικό διαφορικό ή απλά τις μερικές παραγώγους ως προς t, s.

(%i6) `kill(all);`

(%o0) *done*

(%i1) `f(x,y,z):=x^2+y^2+z^2-1;`

(%o1) $f(x,y,z):=x^2+y^2+z^2-1$

(%i2) `x(t):=t^2+2*s^3;y(t):=t^3-5*s^7;z(t):=t^4+6*s^2;`

(%o2) $x(t):=t^2+2\,s^3$

(%o3) $y(t):=t^3-5\,s^7$

(%o4) $z(t):=t^4+6\,s^2$

(%i5) `diff(f(x(t),y(t),z(t)));`

(%o5) $$\left(8\,t^3\left(t^4+6\,s^2\right)+6\,t^2\left(t^3-5\,s^7\right)+4\,t\left(t^2+2\,s^3\right)\right)\mathrm{del}(t)+\left(24\,s\left(t^4+6\,s^2\right)-70\,s^6\left(t^3-5\,s^7\right)+12\,s^2\left(t^2+2\,s^3\right)\right)\mathrm{del}(s)$$

(%i6) `'diff(f(x(t),y(t),z(t)),t)=diff(f(x(t),y(t),z(t)),t);`

(%o6) $$\frac{d}{d\,t}\left(\left(t^4+6\,s^2\right)^2+\left(t^3-5\,s^7\right)^2+\left(t^2+2\,s^3\right)^2-1\right)=8\,t^3\left(t^4+6\,s^2\right)+6\,t^2\left(t^3-5\,s^7\right)+4\,t\left(t^2+2\,s^3\right)$$

(%i7) `'diff(f(x(t),y(t),z(t)),s)=diff(f(x(t),y(t),z(t)),s);`

(%o7) $$\frac{d}{d\,s}\left(\left(t^4+6\,s^2\right)^2+\left(t^3-5\,s^7\right)^2+\left(t^2+2\,s^3\right)^2-1\right)=24\,s\left(t^4+6\,s^2\right)-70\,s^6\left(t^3-5\,s^7\right)+12\,s^2\left(t^2+2\,s^3\right)$$

Σχήμα 7: Σύνθετη παραγώγιση συνάρτησης πολλών μεταβλητών

Παράδειγμα 4.1. *Δίνουμε ένα παράδειγμα μιας εξαιρετικά επίπονης παραγώγισης, εάν την κάνουμε χωρίς κάποιο CAS. Έστω η συνάρτηση:*

$$f(x) = \frac{x^5 + 7x + 5}{x^8 + x^3 + x^2 + 1}$$

Θα υπολογίσουμε την παράγωγό της ως προς x.

```
(%i1) f(x):=(x^5+7*x+5)/(x^8+x^3+x^2+1);
```

(%o1) $f(x):=\frac{x^5+7x+5}{x^8+x^3+x^2+1}$

```
(%i2) diff(f(x),x);
```

(%o2) $\frac{5x^4+7}{x^8+x^3+x^2+1}-\frac{(x^5+7x+5)(8x^7+3x^2+2x)}{(x^8+x^3+x^2+1)^2}$

```
(%i3) factor(%);
```

(%o3) $-\frac{3x^{12}+49x^8+38x^7-3x^6-5x^4+14x^3+22x^2+10x-7}{(x^8+x^3+x^2+1)^2}$

Σχήμα 8: Μία 'κουραστική' παραγώγιση

Παράδειγμα 4.2. *Έστω η συνάρτηση παραγωγής Cobb-Douglas:*

$$Q(K, L) = 100\, K^{\frac{3}{5}}\, L^{\frac{2}{5}}$$

Να υπολογίσετε την κλίση των ισοϋψών καμπυλών αυτής (των καμπυλών ισοπαραγωγής)

```
(%i1) Q(K,L):=100*K^(3/5)*L^(2/5);
```

(%o1) $Q(K,L):=100\,K^{3/5}\,L^{2/5}$

```
(%i2) diff(Q(K,L))=0;
```

(%o2) $\frac{40\,K^{3/5}\,\mathrm{del}(L)}{L^{3/5}}+\frac{60\,L^{2/5}\,\mathrm{del}(K)}{K^{2/5}}=0$

```
(%i3) solve(%,del(K))[1];
```

(%o3) $\mathrm{del}(K)=-\frac{2\,K\,\mathrm{del}(L)}{3\,L}$

```
(%i4) %/del(L);
subst([del(L)=dL,del(K)=dK],%);
```

(%o4) $\frac{\mathrm{del}(K)}{\mathrm{del}(L)}=-\frac{2\,K}{3\,L}$

(%o5) $\frac{dK}{dL}=-\frac{2\,K}{3\,L}$

Σχήμα 9: Κλίση καμπύλων ίσου προϊόντος για την $Q(K, L) = 100\, K^{\frac{3}{5}}\, L^{\frac{2}{5}}$

5 Ολοκλήρωση

Υπολογίζουμε το αόριστο ολοκλήρωμα μιας συνάρτησης $f(x)$ πληκτρολογώντας απλά την εντολή integrate(f(x),x).

```
(%i1) f(x):=x^4*cos(x);
(%o1) f(x):=x^4 cos(x)
```

```
(%i2) 'integrate(f(x),x)=integrate(f(x),x);
```

$$\int x^4 \cos(x)dx = \left(x^4 - 12x^2 + 24\right)\sin(x) + \left(4x^3 - 24x\right)\cos(x) \quad \text{(\%o2)}$$

Το wxMaxima δίνει πολλές φορές τα πιο κομψά αποτελέσματα ολοκλήρωσης συγκριτικά με τους ‘ επιγόνους ’ αυτού, δηλ. το Mathimatica και το Maple, επειδή διαθέτει πολύ καλούς αλγόριθμους συμβολικής ολοκλήρωσης και απλοποίησης.

Παράδειγμα 5.1. *Υπολογίζουμε το ολοκλήρωμα:*

$$\int sin^3 x\, dx = \int sin^2 x\, sinx\, dx = \int (1 - cos^2 x)\, d(-cosx) = \cdots = \frac{cos^3 x}{3} - cosx$$

και με τα τρία προγράμματα:

```
(%i1) integrate (sin(x)^3, x);
(%o1) cos(x)^3/3 - cos(x)
```

(α′) wxMaxima

```
> int(cos(x)^3,x);
```

$$\frac{1}{3}\cos(x)^2\sin(x) + \frac{2}{3}\sin(x)$$

(β′) Maple

```
In[14]:= Integrate[Sin[x]^3, x]
Out[14]= -3 Cos[x]/4 + 1/12 Cos[3 x]
```

(γ′) Mathimatica

Σχήμα 10: Υπολογισμός του ολοκληρώματος $\int cos^3 x\, dx = \frac{cos^3 x}{3} - cosx$

Διαπιστώνουμε ότι το wxMaxima ακολούθησε τον πιο αποτελεσματικό και έξυπνο τρόπο.

Με την ίδια σχεδόν εντολή υπολογίζουμε και ένα ορισμένο ολοκλήρωμα.

```
(%i1) 'integrate(sqrt(x), x, 0, 1)=integrate(sqrt(x), x, 0, 1);
```

$$\text{(\%o1)} \quad \int_0^1 \sqrt{x}\,dx = \frac{2}{3}$$

Μπορούμε να υπολογίσουμε γενικευμένα ολοκληρώματα με απόλυτη ακρίβεια, όταν αυτά συγκλίνουν, δηλ. δεν απειρίζονται.

```
(%i1) 'integrate(1/(x^2+1),x,0,inf)=integrate(1/(x^2+1),x,0,inf);
```

$$\text{(\%o1)} \quad \int_0^\infty \frac{1}{x^2+1}\,dx = \frac{\pi}{2}$$

```
(%i2) 'integrate((x*sin(%pi*x))/(x^2+2*x+5),x,-inf,inf);
```

$$\text{(\%o2)} \quad \int_{-\infty}^\infty \frac{x\,\sin(\pi\,x)}{x^2+2\,x+5}\,dx$$

```
(%i3) integrate((x*sin(%pi*x))/(x^2+2*x+5),x,-inf,inf);
```

$$\text{(\%o3)} \quad -\pi\,\%e^{-2\,\pi}$$

Τέλος, έχουμε αρκετούς τρόπους να υπολογίσουμε ένα ορισμένο ολοκλήρωμα αριθμητικά.

Παράδειγμα 5.2. *Θα υπολογίσουμε το φαινομενικά 'απλό' ορισμένο ολοκλήρωμα:*

$$\int_0^1 \sqrt{1+x^3}\,dx$$

```
(%i1) 'integrate(sqrt(1+x^3),x,0,1);
```

$$\text{(\%o1)} \quad \int_0^1 \sqrt{x^3+1}\,dx$$

```
(%i2) quad_qags(sqrt(x^3+1), x, 0, 1);
(%o2) [1.111447970532576,1.2339551275581681 10^-14,21,0]
(%i3) %[1];
(%o3) 1.111447970532576
```

Στην λίστα που δόθηκε ως απάντηση ο πρώτος όρος είναι το αποτέλεσμα, ο δεύτερος όρος το αναμενόμενο σφάλμα της αριθμητικής μεθόδου, ο τρίτος όρος είναι ο αριθμός των ολοκληρωμάτων που υπολογίστηκαν και ο τέταρτος όρος σημαίνει ότι δεν υπήρξε κάποιο πρόβλημα.

6 Πίνακες

6.1 Ορισμοί και πράξεις Πινάκων

Εισάγουμε έναν πίνακα κατά γραμμές ([...]) με την εντολή matrix([...],[...],...,[...]).

```
(%i1) A:matrix([1,2,3],[4,5,6],[7,8,9]);
      B:matrix([-1,1,2],[3,1,1],[4,2,1]);
```

(%o1) $\begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{bmatrix}$

(%o2) $\begin{bmatrix} -1 & 1 & 2 \\ 3 & 1 & 1 \\ 4 & 2 & 1 \end{bmatrix}$

Οι πράξεις στοιχείο με στοιχείο είναι οι γνωστές πρόσθεση (+), αφαίρεση (-), πολλαπλασιασμός (*) και διαίρεση (/).

```
(%i3) A+B;A-B;A*B;A/B;
```

(%o3) $\begin{bmatrix} 0 & 3 & 5 \\ 7 & 6 & 7 \\ 11 & 10 & 10 \end{bmatrix}$

(%o4) $\begin{bmatrix} 2 & 1 & 1 \\ 1 & 4 & 5 \\ 3 & 6 & 8 \end{bmatrix}$

(%o5) $\begin{bmatrix} -1 & 2 & 6 \\ 12 & 5 & 6 \\ 28 & 16 & 9 \end{bmatrix}$

(%o6) $\begin{bmatrix} -1 & 2 & \frac{3}{2} \\ \frac{4}{3} & 5 & 6 \\ \frac{7}{4} & 4 & 9 \end{bmatrix}$

Η δύναμη κάθε στοιχείου του πίνακα δίνεται με απλό σύμβολο ύψωσης σε δύναμη (ˆ) , ενώ η δύναμη $A^2 = A \cdot A$, που είναι πολλαπλασιασμός πινάκων δίνεται με διπλό σύμβολο ύψωσης σε δύναμη (ˆˆ).

```
(%i7) A^2;'A.'A=A^^2;
```

$$(\%o7) \quad \begin{bmatrix} 1 & 4 & 9 \\ 16 & 25 & 36 \\ 49 & 64 & 81 \end{bmatrix}$$

$$(\%o8) \quad A^2 = \begin{bmatrix} 30 & 36 & 42 \\ 66 & 81 & 96 \\ 102 & 126 & 150 \end{bmatrix}$$

Ο πολλαπλασιασμός πινάκων γίνεται με το σύμβολο της τελίτσας. Υπενθυμίζουμε ότι βάζοντας τον ευθύ τόνο ′ πριν από μία εντολή αυτή η εντολή αδρανοποιείται, κι έτσι μπορούμε να γράψουμε τον πολλαπλασιασμό πινάκων στην ίδια γραμμή με το αποτέλεσμα της πράξης.

```
(%i9) 'A.'B=A.B;
```

$$(\%o9) \quad A \,.\, B = \begin{bmatrix} 17 & 9 & 7 \\ 35 & 21 & 19 \\ 53 & 33 & 31 \end{bmatrix}$$

```
(%i10) 'B.'A=B.A;
```

$$(\%o10) \quad B \,.\, A = \begin{bmatrix} 17 & 19 & 21 \\ 14 & 19 & 24 \\ 19 & 26 & 33 \end{bmatrix}$$

Η αναστροφή πίνακα γίνεται με την εντολή transpose.

```
(%i16) 'A^T=transpose(A);
```

$$(\%o16) \quad A^T = \begin{bmatrix} 1 & 4 & 7 \\ 2 & 5 & 8 \\ 3 & 6 & 9 \end{bmatrix}$$

Μπορούμε να ορίσουμε υπο-πίνακες από οποιονδήποτε πίνακα.

```
(%i1) A:matrix([1,-2,1,-1],[0,1,3,5],[0,0,0,1],[0,0,2,-2]);
      B:matrix([0,-2,0,1],[0,0,-5,0],[1,0,0,0],[0,1,0,1]);
```

(%o1) $\begin{bmatrix} 1 & -2 & 1 & -1 \\ 0 & 1 & 3 & 5 \\ 0 & 0 & 0 & 1 \\ 0 & 0 & 2 & -2 \end{bmatrix}$

(%o2) $\begin{bmatrix} 0 & -2 & 0 & 1 \\ 0 & 0 & -5 & 0 \\ 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 1 \end{bmatrix}$

```
(%i3) submatrix(1,2,A,1,2);
```

(%o3) $\begin{bmatrix} 0 & 1 \\ 2 & -2 \end{bmatrix}$

```
(%i4) submatrix(1,2,A);
```

(%o4) $\begin{bmatrix} 0 & 0 & 0 & 1 \\ 0 & 0 & 2 & -2 \end{bmatrix}$

```
(%i5) submatrix(A,1,2);
```

(%o5) $\begin{bmatrix} 1 & -1 \\ 3 & 5 \\ 0 & 1 \\ 2 & -2 \end{bmatrix}$

Ο **αντίστροφος** ενός πίνακα δίνεται με την (-1) δύναμη πίνακα ή με την εντολή invert. Π.χ. για τον 4×4 πίνακα A που ορίσαμε προηγουμένως έχουμε:



(α′) ως δύναμη (β′) ως εντολή

Σχήμα 11: Αντίστροφος τετραγωνικού πίνακα

Το γινόμενο **Kronecker** δύο πινάκων $A \otimes B$ γίνεται πολύ απλά με την εντολή kronecker_product (A, B).





Ισχύουν ορισμένες ιδιότητες, όπως η προσεταιριστική και η επιμεριστική ως προς την πρόσθεση:

```
(%i28) K1='kronecker_product('kronecker_product ('A, 'B),'d);
(%o28) K1=kronecker_product(kronecker_product(A,B),d)
```

```
(%i30) K1:kronecker_product(kronecker_product(A,B),d);
```

(%o30)
$$\begin{bmatrix} 4 & 0 & 2 & 4 & 0 & 2 \\ 2 & 0 & 1 & 2 & 0 & 1 \\ 6 & 0 & 3 & 6 & 0 & 3 \\ 6 & 2 & 8 & 6 & 2 & 8 \\ 3 & 1 & 4 & 3 & 1 & 4 \\ 9 & 3 & 12 & 9 & 3 & 12 \\ -8 & 0 & -4 & 16 & 0 & 8 \\ -4 & 0 & -2 & 8 & 0 & 4 \\ -12 & 0 & -6 & 24 & 0 & 12 \\ -12 & -4 & -16 & 24 & 8 & 32 \\ -6 & -2 & -8 & 12 & 4 & 16 \\ -18 & -6 & -24 & 36 & 12 & 48 \end{bmatrix}$$

```
(%i33) K2:kronecker_product(A,kronecker_product(B,d));
```

(%o33)
$$\begin{bmatrix} 4 & 0 & 2 & 4 & 0 & 2 \\ 2 & 0 & 1 & 2 & 0 & 1 \\ 6 & 0 & 3 & 6 & 0 & 3 \\ 6 & 2 & 8 & 6 & 2 & 8 \\ 3 & 1 & 4 & 3 & 1 & 4 \\ 9 & 3 & 12 & 9 & 3 & 12 \\ -8 & 0 & -4 & 16 & 0 & 8 \\ -4 & 0 & -2 & 8 & 0 & 4 \\ -12 & 0 & -6 & 24 & 0 & 12 \\ -12 & -4 & -16 & 24 & 8 & 32 \\ -6 & -2 & -8 & 12 & 4 & 16 \\ -18 & -6 & -24 & 36 & 12 & 48 \end{bmatrix}$$

```
(%i36) K3='kronecker_product('B,('A+'C));
(%o36) K3=kronecker_product(B,C+A)
```

```
(%i38) K3:kronecker_product(B,C+A);
```

(%o38)
$$\begin{bmatrix} 12 & -4 & 0 & 0 & 6 & -2 \\ -10 & 2 & 0 & 0 & -5 & 1 \\ 18 & -6 & 6 & -2 & 24 & -8 \\ -15 & 3 & -5 & 1 & -20 & 4 \end{bmatrix}$$

```
(%i39) K4='kronecker_product('B,'A)+'kronecker_product('B,'C);
(%o39) K4=kronecker_product(B,C)+kronecker_product(B,A)
```

```
(%i40) K4:kronecker_product(B,C)+kronecker_product(B,A);
```

(%o40)
$$\begin{bmatrix} 12 & -4 & 0 & 0 & 6 & -2 \\ -10 & 2 & 0 & 0 & -5 & 1 \\ 18 & -6 & 6 & -2 & 24 & -8 \\ -15 & 3 & -5 & 1 & -20 & 4 \end{bmatrix}$$

Η **ορίζουσα** ενός πίνακα A δίνεται με την εντολή determinant(A). Ο $n \times n$ μοναδιαίος πίνακας δίνεται με την εντολή ident(n). Το **χαρακτηριστικό πολυώνυμο** βρίσκεται και με την εντολή charpoly(A,x) ή πιο αναλυτικά με τον ορισμό.

```
(%i1) A:matrix([1,1,2],[1,2,1],[2,1,1]);
```

$$(\%o1)\quad \begin{bmatrix} 1 & 1 & 2 \\ 1 & 2 & 1 \\ 2 & 1 & 1 \end{bmatrix}$$

```
(%i2) AxI:A-x*ident(3);
```

$$(\%o2)\quad \begin{bmatrix} 1-x & 1 & 2 \\ 1 & 2-x & 1 \\ 2 & 1 & 1-x \end{bmatrix}$$

```
(%i3) p(x):=determinant(AxI);
(%o3) p(x):=determinant(AxI)
(%i4) p(x);
(%o4) x+((1-x)(2-x)-1)(1-x)+2(1-2(2-x))+1
(%i5) determinant (A - diagmatrix (length (A), x));
(%o5) x+((1-x)(2-x)-1)(1-x)+2(1-2(2-x))+1
(%i6) charpoly(A,x);
(%o6) x+((1-x)(2-x)-1)(1-x)+2(1-2(2-x))+1
(%i7) factor(%);
(%o7) -(x-4)(x-1)(x+1)
```

Μπορούμε να υπολογίσουμε τις **ιδιοτιμές** και τα **ιδιοδιανύσματα** του τετραγωνικού πίνακα A:

```
(%i1) A:matrix([1,1,2],[1,2,1],[2,1,1]);
(%o1) [1 1 2]
      [1 2 1]
      [2 1 1]

(%i2) [vals, vecs] : eigenvectors (A);
(%o2) [[[4,-1,1],[1,1,1]],[[[1,1,1]],[[1,0,-1]],[[1,-2,1]]]]

(%i4) for i thru length (vals[1]) do disp (eigenvalue[i] = vals[1][i],
      multpiplicity[i] = vals[2][i], eigenvector[i] = vecs[i][1]);
```

$eigenvalue_1 = 4$

$multpiplicity_1 = 1$

$eigenvector_1 = [1,1,1]$

$eigenvalue_2 = -1$

$multpiplicity_2 = 1$

$eigenvector_2 = [1,0,-1]$

$eigenvalue_3 = 1$

$multpiplicity_3 = 1$

$eigenvector_3 = [1,-2,1]$

```
(%o4) done
```

Επίσης μπορούμε να βρούμε μόνο τις ιδιοτιμές με την πολλαπλότητά τους:

```
(%i12) eigenvalues(A);
(%o12) [[4,-1,1],[1,1,1]]
```

Ας θεωρήσουμε έναν πίνακα που δεν διαγωνοποιείται. Θα υπολογίσουμε την κανονική μορφή **Jordan** του πίνακα:

$$A = \begin{pmatrix} 0 & 0 & 0 & 0 & 4 \\ 1 & 0 & 0 & 0 & -16 \\ 0 & 1 & 0 & 0 & 25 \\ 0 & 0 & 1 & 0 & -19 \\ 0 & 0 & 0 & 1 & 7 \end{pmatrix}$$

αφού ‘φορτώσουμε’ πρώτα το πακέτο `load("diag")`και θα κάνουμε επαλήθευση:

```
(%i1) A:matrix( [0, 0, 0, 0, 4], [1, 0, 0, 0, -16],
      [0, 1, 0, 0, 25], [0, 0, 1, 0, -19],
      [0, 0, 0, 1, 7]);
```

$$(\%o1)\quad \begin{bmatrix} 0 & 0 & 0 & 0 & 4 \\ 1 & 0 & 0 & 0 & -16 \\ 0 & 1 & 0 & 0 & 25 \\ 0 & 0 & 1 & 0 & -19 \\ 0 & 0 & 0 & 1 & 7 \end{bmatrix}$$

```
(%i2) load("diag")$
(%i3) jordan(A);
(%o3) [[2,2],[1,3]]
(%i4) J:dispJordan(jordan(A));
```

$$(\%o4)\quad \begin{bmatrix} 2 & 1 & 0 & 0 & 0 \\ 0 & 2 & 0 & 0 & 0 \\ 0 & 0 & 1 & 1 & 0 \\ 0 & 0 & 0 & 1 & 1 \\ 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$

(α′) Jordan form

```
(%i5) P: ModeMatrix(A,jordan(A));
```

$$(\%o5)\quad \begin{bmatrix} -14 & 1 & \frac{11}{4} & -\frac{1}{4} & 1 \\ 49 & 0 & -\frac{33}{4} & -2 & 0 \\ -63 & -6 & \frac{143}{16} & \frac{75}{16} & 0 \\ 35 & 8 & -\frac{33}{8} & -\frac{49}{16} & -\frac{1}{2} \\ -7 & -3 & \frac{11}{16} & \frac{5}{8} & \frac{3}{16} \end{bmatrix}$$

```
(%i6) 'P^^(-1).'A.'P=P^^(-1).A.P;
```

$$(\%o6)\quad P^{-1} \,.\, A \,.\, P = \begin{bmatrix} 2 & 1 & 0 & 0 & 0 \\ 0 & 2 & 0 & 0 & 0 \\ 0 & 0 & 1 & 1 & 0 \\ 0 & 0 & 0 & 1 & 1 \\ 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$

(β′) Επαλήθευση

Σχήμα 12: Κανονική μορφή Jordan τετραγωνικού πίνακα

Πιο συνοπτικά μπορούμε με μία μόνο εντολή να έχουμε όλα τα αποτελέσματα:

```
(%i7) [J,P]:[dispJordan(jordan(A)),ModeMatrix(A,jordan(A))];
```

$$(\%o7)\quad \left[\begin{bmatrix} 2 & 1 & 0 & 0 & 0 \\ 0 & 2 & 0 & 0 & 0 \\ 0 & 0 & 1 & 1 & 0 \\ 0 & 0 & 0 & 1 & 1 \\ 0 & 0 & 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} -14 & 1 & \frac{11}{4} & -\frac{1}{4} & 1 \\ 49 & 0 & -\frac{33}{4} & -2 & 0 \\ -63 & -6 & \frac{143}{16} & \frac{75}{16} & 0 \\ 35 & 8 & -\frac{33}{8} & -\frac{49}{16} & -\frac{1}{2} \\ -7 & -3 & \frac{11}{16} & \frac{5}{8} & \frac{3}{16} \end{bmatrix}\right]$$

Αξίζει να σημειωθεί ότι με τον τρόπο που μόλις είδαμε μπορούμε να κάνουμε **διαγωνοποίηση** ενός πίνακα A:

```
(%i65) A:matrix([1,1,2],[1,2,1],[2,1,1]);
```

$$(\%o65)\quad \begin{bmatrix} 1 & 1 & 2 \\ 1 & 2 & 1 \\ 2 & 1 & 1 \end{bmatrix}$$

```
(%i66) [D,P]:[dispJordan(jordan(A)),ModeMatrix(A,jordan(A))];
```

$$(\%o66)\quad \left[\begin{bmatrix} 4 & 0 & 0 \\ 0 & -1 & 0 \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 1 & 1 & 1 \\ 1 & 0 & -2 \\ 1 & -1 & 1 \end{bmatrix}\right]$$

```
(%i67) 'A=P.D.P^^(-1);
```

$$(\%o67)\quad A = \begin{bmatrix} 1 & 1 & 2 \\ 1 & 2 & 1 \\ 2 & 1 & 1 \end{bmatrix}$$

Παράδειγμα 6.1. *Έστω ότι έχουμε να λύσουμε το 2×3 σύστημα:*

$$\begin{aligned} x - y + 2z &= 12 \\ 2x + 3y - 2z &= 4 \end{aligned}$$

Υπολογίζουμε τον ψευδοαντίστροφο ενός πίνακα, αφού πρώτα 'φορτώσουμε' το πακέτο Γραμμικής Άλγεβρας linearalgebra του wxMaxima και έπειτα λύνουμε το σύστημα με έναν πολλαπλασιασμό:

```
(%i1) load(linearalgebra)$
(%i2) A:matrix([1,-1,2],[2,3,-2])$
      b:matrix([12],[4])$
(%i4) Ampinv:moore_penrose_pseudoinverse(A);
```

(%o4) $\begin{bmatrix} \frac{27}{77} & \frac{17}{77} \\ -\frac{2}{77} & \frac{13}{77} \\ \frac{24}{77} & -\frac{2}{77} \end{bmatrix}$

```
(%i5) x:Ampinv.b;
```

(%o5) $\begin{bmatrix} \frac{56}{11} \\ \frac{4}{11} \\ \frac{40}{11} \end{bmatrix}$

```
(%i6) 'A.'x=A.x;
```

(%o6) $A \; . \; x = \begin{bmatrix} 12 \\ 4 \end{bmatrix}$

Σχήμα 13: Λύση ελαχίστων τετραγώνων συστήματος 2×3 με ψευδοαντίστροφο

6.2 Ανάλυση Πινάκων (Matrix Decomposition)

Εκτός από την διαγωνοποίηση / "jordanοποίηση" που έχουμε ήδη παρουσιάσει, υπάρχουν πολλές ακόμη αναλύσεις ή παραγοντοποιήσεις πινάκων.

1. **LU** παραγοντοποίηση πίνακα.
   Γίνεται με την εντολή

   ```
   [P,L,U]:get_lu_factors (lu_factor(A));
   ```



Σχήμα 14: **LU** παραγοντοποίηση πίνακα

2. **Cholesky** παραγοντοποίηση θετικά ορισμένου συμμετρικού πίνακα.



Σχήμα 15: **Cholesky** παραγοντοποίηση πίνακα

Με κατάλληλη εφαρμογή της θεωρίας μπορούμε να κάνουμε και οποιαδήποτε άλλη παραγοντοποίηση πινάκων.

Παράδειγμα 6.2. *Να γίνει η ανάλυση ιδιαζουσών τιμών (singular value decomposition - SVD) του πίνακα:*

$$A = \begin{pmatrix} 1 & 2 & 0 \\ 2 & 0 & 2 \end{pmatrix}$$

Υπολογίζουμε διαδοχικά, προσέχοντας να βάζουμε τις ιδιοτιμές στην σωστή θέση κάθε φορά:

```
(%i3) A:matrix([1,2,0],[2,0,2]);
      AAT:A.transpose(A)$
      ATA:transpose(A).A$
      LAAT:eigenvalues(AAT)[1]$
      sigma:sqrt(LAAT)$
      'sigma=sigma;
      LEIG:uniteigenvectors(AAT)[2]$
      LL:makelist(LEIG[i][1],i,1,2)$
      U:transpose(matrix(LL[1],LL[2]))$
      'U=U;
      REIG:uniteigenvectors(ATA)[2]$
      LR:makelist(REIG[i][1],i,1,3)$
      V:transpose(matrix(LR[3],LR[2],LR[1]))$
      'V=V;
      Sigma:matrix([3,0,0],[0,2,0])$
      'Sigma=Sigma;
      'U.'Sigma.'V^T=U.Sigma.transpose(V);
```

(%o3) $\begin{bmatrix} 1 & 2 & 0 \\ 2 & 0 & 2 \end{bmatrix}$

(%o8) $\sigma = [3, 2]$

(%o12) $U = \begin{bmatrix} \frac{1}{\sqrt{5}} & \frac{2}{\sqrt{5}} \\ \frac{2}{\sqrt{5}} & -\frac{1}{\sqrt{5}} \end{bmatrix}$

(%o16) $V = \begin{bmatrix} \frac{\sqrt{5}}{3} & 0 & \frac{2}{3} \\ \frac{2}{3\sqrt{5}} & \frac{2}{\sqrt{5}} & -\frac{1}{3} \\ \frac{4}{3\sqrt{5}} & -\frac{1}{\sqrt{5}} & -\frac{2}{3} \end{bmatrix}$

(%o18) $\Sigma = \begin{bmatrix} 3 & 0 & 0 \\ 0 & 2 & 0 \end{bmatrix}$

(%o19) $U \,.\, \Sigma \,.\, V^T = \begin{bmatrix} 1 & 2 & 0 \\ 2 & 0 & 2 \end{bmatrix}$

Σχήμα 16: Ανάλυση ιδιαζουσών τιμών πίνακα

6.3 Λογισμός Πινάκων

Μπορούμε να ορίσουμε ‘συνάρτηση πίνακα’ και μέσω αυτής να κάνουμε εξαιρετικά επίπονες πράξεις πινάκων, όπως λ.χ. η πράξη e^A ή e^{At}, που χρειάζεται για να λύσουμε ένα σύστημα γραμμικών διαφορικών εξισώσεων με σταθερούς συντελεστές. Ας υποθέσουμε ότι έχουμε τον πίνακα:

$$A = \begin{pmatrix} 0 & 0 & 0 & 0 & 4 \\ 1 & 0 & 0 & 0 & -16 \\ 0 & 1 & 0 & 0 & 25 \\ 0 & 0 & 1 & 0 & -19 \\ 0 & 0 & 0 & 1 & 7 \end{pmatrix}$$

Φορτώνουμε τα πακέτα eigen,diag,facexp και υπολογίζουμε:

```
(%i15) load("eigen")$
       load("diag")$
       load("facexp")$

(%i26) mat_function(exp,t*A)$
       eat1:subst(e,%e,%)$
       eAT:facsum (eat1, e^t, e^(2*t))$
       e^('A*t)=eAT;
```

(%o29)
$$e^{tA} = \begin{bmatrix} 2e^t(t^2+2t+4)+e^{2t}(2t-7) & 2e^t(t^2+4t+6)+4e^{2t}(t-3) & 2e^t(t^2+6t+10)+4e^{2t}(2t-5) & 2e^t(t+4)^2+16e^{2t}(t-2) & 16e^{2t}(2t-3)+2e^t(t+4)(t+6) \\ -2e^t(3t^2+8t+12)-e^{2t}(7t-24) & -2e^t(3t^2+14t+20)-e^{2t}(14t-41) & -2e^t(3t^2+20t+34)-4e^{2t}(7t-17) & -2e^t(3t^2+26t+54)-4e^{2t}(14t-27) & -16e^{2t}(7t-10)-2e^t(t+4)(3t+20) \\ \frac{e^t(13t^2+42t+60)+6e^{2t}(3t-10)}{2} & \frac{e^t(13t^2+68t+102)+6e^{2t}(6t-17)}{2} & \frac{e^t(13t^2+94t+170)+24e^{2t}(3t-7)}{2} & \frac{e^t(13t^2+120t+264)+24e^{2t}(6t-11)}{2} & \frac{e^t(13t^2+146t+384)+96e^{2t}(3t-4)}{2} \\ -e^t(3t^2+11t+16)-e^{2t}(5t-16) & -e^t(3t^2+17t+27)-e^{2t}(10t-27) & -4e^{2t}(5t-11)-e^t(t+4)(3t+11) & -e^t(3t^2+29t+67)-4e^{2t}(10t-17) & -e^t(3t^2+35t+96)-16e^{2t}(5t-6) \\ \frac{e^t(t^2+4t+6)+2e^{2t}(t-3)}{2} & \frac{e^t(t^2+6t+10)+2e^{2t}(2t-5)}{2} & \frac{e^t(t+4)^2+8e^{2t}(t-2)}{2} & \frac{8e^{2t}(2t-3)+e^t(t+4)(t+6)}{2} & \frac{e^t(t^2+12t+34)+32e^{2t}(t-1)}{2} \end{bmatrix}$$

Σχήμα 17: Εκθετικός Πίνακας e^{At}

Υπολογίζουμε μία τετραγωνική ρίζα ενός τετραγωνικού πίνακα A, δηλ. την λύση της εξίσωσης $X^2 = A$, εάν φυσικά υπάρχει:

```
(%i1) load("eigen")$
      load("diag")$
      load("facexp")$
```

```
(%i4) A:matrix([-3,2,2],[-6,5,2],[-7,4,4]);
```

(%o4)
$$\begin{bmatrix} -3 & 2 & 2 \\ -6 & 5 & 2 \\ -7 & 4 & 4 \end{bmatrix}$$

```
(%i5) sa:mat_function(sqrt,A);
      'sa*'sa=radcan(sa.sa);
```

(%o5)
$$\begin{bmatrix} -3^{3/2}+2^{3/2}+2 & 3^{3/2}-2^{5/2}+1 & 2^{3/2}-2 \\ -4\sqrt{3}+2^{3/2}+2 & 4\sqrt{3}-2^{5/2}+1 & 2^{3/2}-2 \\ -5\sqrt{3}+3\sqrt{2}+2 & 5\sqrt{3}-3\,2^{3/2}+1 & 3\sqrt{2}-2 \end{bmatrix}$$

(%o6)
$$sa^2=\begin{bmatrix} -3 & 2 & 2 \\ -6 & 5 & 2 \\ -7 & 4 & 4 \end{bmatrix}$$

Σχήμα 18: Τετραγωνική ρίζα Πίνακα

Μπορούμε να υπολογίσουμε τον λογάριθμο ενός πίνακα, λ.χ. για τον ίδιο πίνακα A του οποίου υπολογίσαμε την τετραγωνική ρίζα, ο λογάριθμος είναι:

```
(%i27) la:mat_function(log,A);
       logcontract(%);
       e^logA=mat_function(exp,la);
```

(%o27) $\begin{bmatrix} 2\log(2)-3\log(3) & 3\log(3)-4\log(2) & 2\log(2) \\ 2\log(2)-4\log(3) & 4\log(3)-4\log(2) & 2\log(2) \\ 3\log(2)-5\log(3) & 5\log(3)-6\log(2) & 3\log(2) \end{bmatrix}$

(%o28) $\begin{bmatrix} \log\left(\frac{4}{27}\right) & \log\left(\frac{27}{16}\right) & \log(4) \\ \log\left(\frac{4}{81}\right) & \log\left(\frac{81}{16}\right) & \log(4) \\ \log\left(\frac{8}{243}\right) & \log\left(\frac{243}{64}\right) & \log(8) \end{bmatrix}$

(%o29) $e^{logA}=\begin{bmatrix} -3 & 2 & 2 \\ -6 & 5 & 2 \\ -7 & 4 & 4 \end{bmatrix}$

Σχήμα 19: Φυσικός λογάριθμος Πίνακα

6.4 Γραμμική Αλγεβρα

Υπολογίζουμε μία **βάση** του διανυσματικού χώρου που παράγεται από τις στήλες ενός πίνακα:

```
(%i1) A:matrix([1,2,-2,0,2,0,0],[1,3,-5,-2,4,-3,-1],
      [0,0,1,2,0,2,1],[1,3,0,1,2,7,3]);
```

(%o1) $\begin{bmatrix} 1 & 2 & -2 & 0 & 2 & 0 & 0 \\ 1 & 3 & -5 & -2 & 4 & -3 & -1 \\ 0 & 0 & 1 & 2 & 0 & 2 & 1 \\ 1 & 3 & 0 & 1 & 2 & 7 & 3 \end{bmatrix}$

```
(%i2) columnspace(A);
```

(%o2) $\mathrm{span}\left(\begin{bmatrix} -2 \\ -5 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ -2 \\ 2 \\ 1 \end{bmatrix}, \begin{bmatrix} 1 \\ 1 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} 2 \\ 3 \\ 0 \\ 3 \end{bmatrix}\right)$

Σχήμα 20: Βάση του χώρου στηλών ενός Πίνακα

Βρίσκουμε τον **πυρήνα** του ίδιου πίνακα:



Σχήμα 21: Πυρήνας ενός Πίνακα

Επομένως ο βαθμός του πίνακα A θα είναι $rank(A) = 4$, όπως και βρίσκουμε:



Σχήμα 22: Βαθμός Πίνακα

Μπορούμε να βρούμε την κλιμακωτή μορφή του A κάνοντας απαλοιφή Gauss και να επαληθεύσουμε τα ανωτέρω αποτελέσματα:



Σχήμα 23: Κλιμακωτή μορφή Πίνακα

7 Διαφορικές Εξισώσεις

Λύνουμε την συνήθη διαφορική εξίσωση:

$$\frac{d^2y}{d\,x^2}+4\,\frac{dy}{d\,x}+4\,y=3\,x$$

```
(%i5) deq:'diff(y(x),x,2)+4*'diff(y(x),x)+4*y(x)=3*x;
```

(%o5) $\frac{d^2}{d\,x^2}\,y(x)+4\left(\frac{d}{d\,x}\,y(x)\right)+4\,y(x)=3\,x$

```
(%i6) desolve(deq,y(x));
```

(%o6) $y(x)=x\,\%e^{-2\,x}\left(\left.\frac{d}{d\,x}\,y(x)\right|_{x=0}\right)+2\,y(0)\,x\,\%e^{-2\,x}+\frac{3\,x\,\%e^{-2\,x}}{4}+\frac{(4\,y(0)+3)\,\%e^{-2\,x}}{4}+\frac{3\,x}{4}-\frac{3}{4}$

Σχήμα 24: Λύση συνήθους διαφορικής εξίσωσης

Μπορούμε να λύσουμε το Πρόβλημα Αρχικών Τιμών:

$$\frac{d^2y}{d\,x^2}+4\,\frac{dy}{d\,x}+4\,y=3\,x,y(0)=1,y^{'}(0)=-1$$

```
(%i1) deq:'diff(y(x),x,2)+4*'diff(y(x),x)+4*y(x)=3*x;
```

(%o1) $\frac{d^2}{d\,x^2}\,y(x)+4\left(\frac{d}{d\,x}\,y(x)\right)+4\,y(x)=3\,x$

```
(%i2) initial_conditions:
      [atvalue(y(x),x=0,1),atvalue('diff(y(x),x),x=0,-1)];
(%o2) [1,-1]
(%i3) desolve(deq,y(x));
```

(%o3) $y(x)=\frac{7\,x\,\%e^{-2\,x}}{4}+\frac{7\,\%e^{-2\,x}}{4}+\frac{3\,x}{4}-\frac{3}{4}$

Σχήμα 25: Λύση Π.Α.Τ.

8 Γραφικά

Κάνουμε με τρεις τρόπους το γράφημα της συνάρτησης μιας μεταβλητής:

$$\mathrm{f}\left(x\right) = \frac{1}{1+x^2}$$

```
(%i1) f(x):=1/(1+x^2);
```

(%o1) $\mathrm{f}(x) := \dfrac{1}{1+x^2}$

```
(%i2) wxplot2d(f(x),[x,-3,3]);
```

(%t2)



(%o2)

Σχήμα 26: Γράφημα inline 2 διαστάσεων



Σχήμα 27: Γράφημα xmaxima 2 διαστάσεων



Σχήμα 28: Γράφημα gnuplot 2 διαστάσεων

Και ένα γράφημα μιας συνάρτησης δύο μεταβλητών:

$$\mathrm{f}\left(x,y\right)=\cos\left(-x^2+\frac{y^3}{4}\right)$$

```
(%i29) f(x,y):=cos (-x^2 + y^3/4);
```

$$\text{(\%o29)}\quad \mathrm{f}(x,y):=\cos\left(-x^2+\frac{y^3}{4}\right)$$

```
(%i31) plot3d (cos (-x^2 + y^3/4), [x, -4, 4], [y, -4, 4],
           [mesh_lines_color, false], [elevation, 16], [azimuth, 32],
           [colorbox, true], [grid, 150, 150])$
```



Σχήμα 29: Γράφημα 3 διαστάσεων

9 Τελικά

Οι ανωτέρω πολύ σύντομες σημειώσεις δεν εξαντλούν σε καμία περίπτωση τον πλούτο και την έκταση εφαρμογών του wxMaxima. Γράφτηκαν για να κεντρίσουν το ενδιαφέρον της ακαδημαϊκής κοινότητας, ώστε να ασχοληθεί και με κάτι μη εμπορικό, αλλά αξιόπιστο και πλήρως λειτουργικό. Εάν τις βρήκατε ενδιαφέρουσες, μην διστάσετε να επικοινωνήσετε με τον γράφοντα[2] για να εκφράσετε την άποψή σας, ώστε να εμπλουτιστούν με θέματα και εφαρμογές που σας ενδιαφέρουν.

Στόχος μας είναι να αναπτυχθεί και στην Ελλάδα μία κοινότητα χρηστών του wxMaxima, η οποία θα αναπτύξει εφαρμογές χρήσιμες για τα μαθήματα που διδάσκονται στα Ελληνικά Πανεπιστήμια, αλλά και στην πραγματοποίηση βασικής έρευνας στα Ερευνητικά Ιδρύματα γενικότερα.

[2] Δημήτριος Θ. Χριστόπουλος, dchristop@econ.uoa.gr